;     Функционално програмиране.

;     Упражнение N8

;     Дефиниции на функции за работа със списъци.

;Задача 1. Да се напише функция (union x y), която връща

;обединението на елементите на множествата x и y.

(define (union x y)

  (cond ((null? x) y)

        ((member (car x) y) (union (cdr x) y))

        (else (cons (car x) (union (cdr x) y)))))

;Задача 2. Да се напише функция (sect x y), която връща сечението

;на множествата x и y.

(define (sect x y)

  (cond ((null? x) ())

        ((null? y) ())

        ((member (car x) y) (cons (car x) (sect (cdr x) y)))

        (else (sect (cdr x) y))))

;Задача 3. Да се напише функция (diff x y), която връща разликата

;на множествата x и y.

(define (diff x y)

  (cond ((null? x) ())

        ((null? y) x)

        ((not (member (car x) y)) (cons (car x) (diff (cdr x) y)))

        (else (diff (cdr x) y))))

;Задача 4. Да се напише функция (common x y), която проверява дали

;множествата х и y имат общ елемент.

(define (common x y)

  (cond ((null? x) ())

        ((memq (car x) y) #T)

        (else (common (cdr x) y))))

;Задача 5. Да се напише функция (equal-sets m n), която проверява

;дали m и n се състоят от едни и същи елементи.

(define (equal-sets m n)

  (define (subset a b)

    (if (null? a)

        #t

        (and (member (car a) b) (subset (cdr a) b))))

  (and (subset m n) (subset n m)))

;Задача 6. Да се напише функция (set m), проверява дали m е

;множество (елементите му са атоми и няма повтарящи се елементи).
(require (lib "compat.ss"))

(define (set m)

  (cond ((null? m) #T)

        ((or (not (atom? (car m))) (memq (car m) (cdr m))) #F)

        (else (set (cdr m)))))

;Задача 7. Да се напише функция (rem-dubl l), която връща списъка

;l, от който са премахнати всички повтарящи се елементи в него.

(define (rem-dubl l)

  (if (null? l)

      ()
      (cons (car l) (rem-dubl (delete-all (car l) (cdr l))))))

;Задача 8. Да се напише функция (num-el elem list), която
;проверява колко пъти elem се съдържа в list като елемент.

(define (num-el elem list)

  (cond ((null? list) 0)

        ((equal? elem (car list)) (+ 1 (num-el elem (cdr list))))

        (else (num-el elem (cdr list)))))

;Задача 9. Да се напише функция (ord x), която проверява дали

;елементите на списъка x (числа) са подредени във възходящ ред.

(define (ord x)

  (cond ((null? x) #T)

        ((pair? (cdr x)) (and (<= (car x) (cadr x)) (ord (cdr x))))

        (else #T)))

;Задача 10. Да се напише функция (insert elem list), която вмъква

;число elem в подредения списък list.

(define (insert elem list)

  (cond ((null? list) (cons elem ()))

        ((< elem (car list)) (cons elem list))

        (else (cons (car list) (insert elem (cdr list))))))

;Задача 11. Да се напише функция, която намира максималния елемент

;на даден списък от числа.

(define (maxel l)

  (if (null? (cdr l))

      (car l)

      (max (car l) (maxel (cdr l)))))

;     Дефиниции на функции за работа със списъци на произволно

;ниво на вложение.

;Задача 1. Да се дефинира функция, която връща списък от всички

;атоми, участващи в даден списък като елементи на кое да е ниво.

(define (flatten l)

  (cond ((null? l) ())

        ((atom? l) (list l))

        (else (append (flatten (car l)) (flatten (cdr l)))) ))

;Задача 2. Да се дефинира функция, която намира максималната

;дълбочина на вложение в даден списък.

(define (depth l)

  (cond ((null? l) 1)

        ((atom? l) 0)

        (else (max (1+ (depth (car l)))

                   (depth (cdr l)))) ))

;Задача 3. Да се дефинира функция, която проверява дали даден

;атом a се среща в даден списък l като елемент на произволно ниво

;на вложение.

(define (super-member a l)

  (cond ((eqv? a l) #t)
        ((atom? l) #f)
        (else (or (super-member a (car l))

                  (super-member a (cdr l)))) ))

;Задача 4. Да се дефинира функция, която замества всяко срещане на

;s-израза s1 в списъка l c s-израза s2, като замяната се извършва

;на всички нива на вложение.

(define (supersubst s1 s2 l)

  (cond ((null? l) ())

        ((equal? (car l) s1) (cons s2 (supersubst s1 s2 (cdr l))))

        ((atom? (car l)) (cons (car l) (supersubst s1 s2 (cdr l))))

        (else (cons (supersubst s1 s2 (car l))

                    (supersubst s1 s2 (cdr l))))))

;Задача 5. Да се дефинира функция (rotate-right l n), която

;завърта надясно списъка l на n позиции, и функция 
;(rotate-left l n), която завърта наляво списъка l на n позиции. ;Като се използват тези две функции, да се напише функция 
;(rotate-list l n direction), която завърта списъка l на n позиции 
;в посока direction.
(require (lib "list.ss"))

(define (rotate-right l n)

  (if (zero? n)

      l

      (rotate-right (append (last-pair l)

                            (reverse (cdr (reverse l))))

                    (- n 1))))

(define (rotate-left l n)

  (if (zero? n)

      l

      (rotate-left (append (cdr l)

                           (list (car l)))

                   (- n 1))))

(define (rotate-list l n direction)

  (cond ((eq? direction 'left) (rotate-left l n))

        ((eq? direction 'right) (rotate-right l n))

        (else l)))

;Задача 6. Да се дефинира функция с аргумент списък, която брои

;отварящите и затварящите скоби на списъка. 
;'() -> 2; '((а b) с) -> 4
;'(((a () b) c) () ((d) e)) -> 14

(define (countpar l)

  (cond ((null? l) 2)

        ((atom? l) 0)

        (else (+ (countpar (car l))

                 (countpar (cdr l))))))

;Задача 7. Да се дефинира функция с аргумент непразен списък, която

;връща атома, който е най-вляво в списъка. '((a b) (c (d e))) -> a

;'((((c ((e f) g) h)))) -> c; '(() a) -> ()

(define (leftmost l)

  (cond ((atom? l) l)

        (else (leftmost (car l)))))

;Задача 8. Да се дефинира функция с аргумент непразен списък, която

;връща атома, който е най-вдясно в списъка. '(a ()) -> ()

;'((((((b (c))))))) -> c 
;'((a b) (d (c d (f (g h) i) m n) u) v) -> v

(define (rightmost l)

  (cond ((atom? l) l)

        ((null? (cdr l)) (rightmost (car l)))

        (else (rightmost (cdr l)))))

